

Ingeniería del Software II

8 - El lenguaje de especificación Alloy

El lenguaje de especificación Alloy

- Alloy es un **lenguaje liviano (lightweight) de modelado** para la especificación y diseño de software.
- Puede **analizarse** de manera automática con el Alloy Analyzer.
- Provee un **visualizador** que ayuda a comprender las soluciones y los contraejemplos que el analizador encuentre.
- **Alloy = lógica de primer orden + álgebra de relaciones.**
 - el lenguaje subyacente de Alloy es la **lógica relacional**.
- A diferencia del álgebra de relaciones, Alloy posee relaciones de cualquier aridad.

Sintaxis

Signaturas

- Las signaturas son la base del lenguaje Alloy.
- Permiten denotar dominios.
- **Ejemplo:** Especificación de un sistema de memorias.
 - Primero necesitamos dos dominios:
 1. direcciones (de memoria);
 2. datos (a almacenarse en direcciones de memoria).
 - En Alloy:

```
sig Addr {}  
sig Data {}
```

Sintaxis

Signaturas

- Las signaturas son la base del lenguaje Alloy.
- Permiten denotar dominios.

- Ejemplo:** Especificación de un sistema de

- Primero necesitamos dos dominios:

1. direcciones (de memoria);
2. datos (a almacenarse en direcciones de memoria).

- En Alloy:

```
sig Addr, Data {}
```

Estas signaturas son **básicas**, ya que no tienen estructura interna.

Alternativamente

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener estructuras complejas.
- Pueden incluir "**campos**":

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}
```

- El ejemplo describe una memoria que
 1. posee un espacio de direcciones (**addrs**); y
 2. cada dirección de memoria puede albergar un dato (**map**).

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener est
- Pueden incluir "**campos**":

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}
```

"Addr -> Data"

denota una relación binaria. Es lo que usualmente denotamos con

Addr×Data.

- El ejemplo describe una memoria que
 1. posee un espacio de direcciones (**addrs**); y
 2. cada dirección de memoria puede albergar un dato (**map**).

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener est
- Pueden incluir "**campos**":

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}
```

¡¡Pero entonces puede ser inconsistente!!

Es lo
on

- El ejemplo describe una memoria que
 1. posee un espacio de direcciones (**addrs**); y
 2. cada dirección de memoria puede albergar un dato (**map**).

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener este tipo de expresiones
- Pueden incluir "**campos**":

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}
```

"... -> Data"
Es lo
on
¡¡Pero entonces puede ser
inconsistente!!

- El ejemplo describe una memoria que:

1. posee un espacio de direcciones

2. cada dirección de memoria

Nada asegura que **map** sea una función, ni que su dominio sea exactamente las direcciones que la memoria puede direccionar (i.e. que **map** sea total en **addrs**).

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener estructuras complejas.
- Pueden incluir “campos” y propiedades:

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}{  
  ~map.map in iden  
  map.Data = addrs  
}
```

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener
- Pueden incluir "**campos**" y

¡Pero esto lo podemos escribir más fácil!

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}{  
  ~map.map in iden  
  map.Data = addrs  
}
```

map es una función
(lo vimos la clase pasada)

map es total.

Notar el uso del 'join' (.)
entre una relación binaria y un
conjunto ("relación unaria")

Sintaxis

Signaturas Complejas

- Las signaturas pueden tener estructuras complejas.
- Pueden incluir "campos" y propiedades.
- Alloy provee un lenguaje rico para definir multiplicidades de conjuntos y relaciones:

```
sig Memory {  
  addrs: set Addr  
  map: addrs -> one Data  
}
```

Sintaxis

Extensión de firmas

- Pueden definirse tipos más específicos de firmas extendiendo otras firmas existentes
- Ejemplo: un sistema de memoria de una computadora consiste de la memoria principal (**MainMemory**) y la memoria cache (**Cache**).

```
sig MainMemory extends Memory { }  
  
sig Cache extends Memory {  
    dirty: set addr  
}
```

Sintaxis

Extensión de firmas

- Pueden definirse tipos más específicos de firmas extendiendo otras firmas existentes
- Ejemplo: un sistema de memoria de una computadora consiste de la memoria principal (**MainMemory**) y la memoria cache (**Cache**).

```
sig MainMemory extends Memory { }  
  
sig Cache extends Memory {  
    dirty: set addr  
}
```

La memoria cache, además, debe llevar un registro de los datos que fueron modificados pero aún no copiados en la memoria

Sintaxis

Completando el sistema del ejemplo

- El sistema completo podría quedar definido como:

```
sig System {  
    cache: Cache  
    main: MainMemory  
}
```

- La intención de esta signatura es describir un sistema (de memoria) como una estructura compuesta por una memoria principal (campo de tipo **MainMemory**) y una cache (campo de tipo **Cache**).

Sintaxis

Operaciones en un modelo Alloy

- Las firmas permiten describir los dominios y sus estructuras.
- Además se pueden especificar predicados sobre estos dominios.
- Los predicados permiten, en particular, definir operaciones que relacionan el estado previo a la aplicación de la operación con el estado posterior correspondiente.

```
pred Write [m_i, m_o: Memory, d: Data, a: Addr] {  
    m_o.map = m_i.map ++ (a -> d)  
}
```

- **Write** modifica una memoria sobrescribiendo el valor asociado a una dirección **a** con el dato **d**.

Sintaxis

Operaciones en un modelo Alloy

Este predicado no está directamente asociado a ninguna signatura.

Permiten describir los dominios y sus estructuras. Permiten especificar predicados sobre estos dominios. Permiten, en particular, definir operaciones que relacionan un estado previo a la aplicación de la operación con el estado posterior correspondiente.

```
pred Write [m_i, m_o: Memory, d: Data, a: Addr] {  
    m_o.map = m_i.map ++ (a -> d)  
}
```

- **Write** modifica una memoria sobrescribiendo el valor asociado a una dirección **a** con el dato **d**.

Sintaxis

Hechos (facts)

- Las especificaciones de la estructura de un sistema que se pueden lograr usando sólo firmas son limitadas.
- Usualmente, las firmas deben complementarse con restricciones estructurales adicionales que permitan describir propiedades o restricciones generales.
- Por ejemplo, podríamos agregar una restricción que asegure que la memoria caché no direcciona fuera de la memoria principal:

```
fact {  
    all s:System | s.cache.addr in s.main.addr  
}
```

Sintaxis

Hechos (facts) (cont.)

Escribir:

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}{  
  ~map.map in iden  
  map.Data = addrs  
}
```

Es lo mismo
que escribir:

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}
```

...pero esto
es más
complicado!

```
fact {  
  all m:Memory | ~(m.map).(m.map) in iden  
  all m:Memory | m.map.Data = m.addrs  
}
```

Sintaxis

Un **fact** de signatura es, de hecho, un invariante de la signatura.

Hechos (facts) (cont.)

Escribir:

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}{  
  ~map.map in iden  
  map.Data = addrs  
}
```

Es lo mismo
que escribir:

```
sig Memory {  
  addrs: set Addr  
  map: Addr -> Data  
}
```

...pero esto
es más
complicado!

```
fact {  
  all m:Memory | ~(m.map).(m.map) in iden  
  all m:Memory | m.map.Data = m.addrs  
}
```

Sintaxis

Aserciones

- Todos los elementos de Alloy descritos hasta el momento forman parte de la descripción de los modelos.
- Las aserciones, en cambio, son enunciados que queremos comprobar si son o no propiedades del sistema especificado.
- Por ejemplo, podríamos querer saber si en el sistema de memorias se cumple la siguiente propiedad :

```
assert {  
    all s: System |  
        all a: s.cache.addrs-s.cache.dirty |  
            s.cache.map[a] = s.main.map[a]  
}
```

Sintaxis

Aserciones

- Todos los elementos de Alloy descritos hasta el momento forman parte de la descripción de los modelos.
- Las aserciones, en cambio, son enunciados que queremos comprobar si son o no propiedades del sistema especificado.
- Por ejemplo, podríamos querer saber si en el sistema de memorias se cumple la siguiente propiedad :

```
assert {  
    all s: System |  
        all a: s.cache.addrs-s.cache.dirty |  
            s.cache.map[a] = s.main.map[a]  
}
```

¿Qué quiere decir?

Sintaxis

Hechos, Predicados y Aserciones

- Un hecho define una restricción que es siempre verdadera.
- Por lo tanto, un hecho limita el dominio de la especificación.
- Un predicado sólo será evaluado cuando se invoque.
- Un predicado puede invocarse desde un hecho, una aserción, una función u otro predicado.
- Una aserción es una restricción que se espera que se deduzca de los hechos del modelo.
- Por lo tanto, Alloy Analyzer verifica la validez de una aserción en un modelo.
- Si una aserción no es válida, o bien ésta expuso un error del modelo, o bien la aserción no estuvo bien formulada.

Características de Alloy

- El lenguaje es **mínimo**, no incluye ni siquiera tipos básicos.
- El lenguaje es **relacional**, con una sintaxis simple y elegante basada en operadores relacionales tales como composición, unión, complemento, etc.
- El lenguaje es **expresivo**, permitiendo cuantificación similar a la cuantificación sobre individuos, pero con una semántica relacional, y clausura transitiva.
- El lenguaje es **fácil de usar**, con construcciones que recuerdan elementos de orientación a objetos, tales como extensión de signatures y la notación de punto para acceso a servicios (pero el lenguaje no es orientado a objetos).

Características de ,

En realidad, tiene **Int** "built in" y bibliotecas (ej: orden total)

- El lenguaje es **mínimo**, no incluye ni siquiera tipos básicos.
- El lenguaje es **relacional**, con una sintaxis simple y elegante basada en operadores relacionales tales como composición, unión, complemento, etc.
- El lenguaje es **expresivo**, permitiendo cuantificación similar a la cuantificación sobre individuos, pero con una semántica relacional, y clausura transitiva.
- El lenguaje es **fácil de usar**, con construcciones que recuerdan elementos de orientación a objetos, tales como extensión de signatures y la notación de punto para acceso a servicios (pero el lenguaje no es orientado a objetos).

Características de Alloy

- el lenguaje hace que las especificaciones sean **analizables automáticamente**, usando técnicas de SAT solving (para validación, pues el lenguaje es de primer orden)
- Alloy trabaja bajo la “**small scope hypothesis**” (hipótesis de dominio pequeño):

La mayoría de los errores tienen contraejemplos pequeños.

- i.e. los sistemas que fallan en una gran instancia, casi seguro falla también en una instancia pequeña con propiedades similares, aún si las pequeñas instancias no ocurren en la práctica.

Alloy como Constraint Solver

- Alloy es una herramienta versátil, que puede utilizarse con diferentes finalidades.
- Ya hemos utilizado Alloy para realizar una especificación y verificar propiedades en ella.
- Alloy también puede utilizarse como constraint solver, i.e., para la resolución de problemas con restricciones.

Alloy como Constraint Solver

Ejemplo: El problema del granjero que debe cruzar un río

- Un granjero se encuentra a orillas de un río llevando un zorro, una gallina y una bolsa de maíz.
- Necesita cruzar el río con el zorro, la gallina y la bolsa de maíz, pero sólo cuenta con un pequeño bote en el que puede llevar a lo sumo una carga. (Por supuesto que el único que sabe remar es el granjero.)
- El granjero no debe dejar solo al zorro con la gallina porque se la puede comer. Por las mismas razones, tampoco puede dejar sola a la gallina junto al maíz.
- ¿Cómo puede hacer para cruzar toda su carga sin que alguno de los animales se haga un festín?
- Podemos utilizar Alloy y el Alloy Analyzer para resolver este problema.

Alloy como Constraint Solver

- Primero modelamos los “objetos” que son el dominio del problema mediante firmas.

```
abstract sig Object {  
    eats: set Object  
}
```

```
one sig Farmer, Fox, Chicken, Grain  
    extends Object {}
```

Allo

La signature **Object** es "abstracta". esto significa que no contendrá elementos a menos que los declaremos explícitamente mediante extensiones.

- Primero m del problema el dominio

```
abstract sig Object {  
    eats: set Object  
}
```

```
one sig Farmer, Fox, Chicken, Grain  
    extends Object {}
```

La multiplicidad **one** garantiza que cada una de estas signatures contendrá exactamente un elemento.

Alloy como Constraint Solver

- La especificación del dominio del problema necesita complementarse con un hecho que explicita quién puede comer qué.

```
fact eating {  
    eats = Fox -> Chicken + Chicken -> Grain  
}
```

Alloy como Co

- La especificación del dominio complementarse con un hecho qué.

Observar que el "campo" **eats** de la signature **Object** es en realidad una relación **Object -> Object** encubierta comer

```
fact eating {  
    eats = Fox -> Chicken + Chicken -> Grain  
}
```

Alloy como Co

Observar que el
"campo" **eats** de la signature
Object es en realidad una relación
Object -> Object
encubierta

- La especificación del dominio complementarse con un hecho qué.

```
fact eating {  
    eats = Fox -> Chicken + Chicken -> Grain  
}
```

- Esta es una forma simple de expresarlo, pero no la única. Otro forma:

```
fact eating {  
    Fox.eats = Chicken  
    Chicken.eats = Grain  
    no Grain.eats  
    no Farmer.eats  
}
```


Alloy como Constraint Solver

Cambios de estado

- Los posibles traslados del bote lo podemos modelar como cambios de estado.
- Para ello necesitamos indicar cómo está compuesto el estado.
- En nuestro caso debemos indicar qué objetos hay en cada lado del río

```
sig State {  
    near: set Object,  
    far: set Object,  
}
```

Alloy como Constraint Solver

Cambios de estado (cont.)

- Para poder identificar la secuencia de los movimientos vamos a necesitar introducir “trazas”.
- Una traza es una secuencia ordenada de estados.
- Alloy provee una definición de orden en un módulo predefinido

```
open util/ordering[State]
```

- Podemos caracterizar el estado inicial de la siguiente manera:

```
fact initialState {  
    let s0 = first[] |  
        s0.near = Object && no s0.far  
}
```

Alloy como Constraint Solver

Cambios de estado (cont.)

- Para poder identificar la secuencia de los movimientos vamos a necesitar introducir “trazas”.
- Una traza es una secuencia ordenada de estados.
- Alloy provee una definición de `ordering` predefinido

Notar que el módulo **ordering** es paramétrico.

```
open util/ordering[State]
```

- Podemos caracterizar el estado inicial

```
fact initialState {  
    let s0 = first[] |  
        s0.near = Object && no s0.far  
}
```

first
es una función del módulo **ordering**. También lo son **next** y **last**.

Alloy como Constraint Solver

Cambios de estado (cont.)

- Debemos especificar ahora como se modifica el estado en un paso, i.e., la transición.
- Para ello utilizaremos un predicado auxiliar que especifica el cruce del río.

Alloy como Constraint Solver

Cambios de estado (cont.)

- Debemos especificar ahora como se modifica el estado en un paso, i.e., la transición.
- Para ello utilizaremos un predicado auxiliar que especifica el cruce del río.

```
pred crossRiver [from_i, from_o, to_i, to_o: set Object ] {  
    ( from_o = from_i - Farmer &&  
      to_o = to_i - to_i.eats + Farmer )  
    ||  
    ( some item: from_i - Farmer |  
      from_o = from_i - Farmer - item &&  
      to_o = to_i - to_i.eats + Farmer + item  
    )  
}
```

Alloy como Constraint Solver

Cambios de estado (cont.)

- Usando **crossRiver** podemos definir las transiciones del sistema como sigue.

```
fact stateTransition {  
    all s1: State, s2: next[s1] |  
        ( Farmer in s.near =>  
            crossRiver[s1.near,s2.near,s1.far,s2.far] )  
        &&  
        ( Farmer in s.far =>  
            crossRiver[s1.far,s2.far,s1.near,s2.near ] )  
}
```

Alloy como Constraint Solver

Cambios de estado (cont.)

- Usando **crossRiver** poder del sistema como sigue.

Notar que este **fact** está definiendo la relación de orden a través de la función **next**.

```
fact stateTransition {  
  all s1: State, s2: next[s1] |  
    ( Farmer in s.near =>  
      crossRiver[s1.near,s2.near,s1.far,s2.far] )  
  &&  
  ( Farmer in s.far =>  
    crossRiver[s1.far,s2.far,s1.near,s2.near ] )  
}
```

Alloy como Constraint Solver

Cambios de estado (cont.)

- Usando un predicado que especifique cual es el estado final deseado, podemos utilizar Alloy para que busque la solución:

```
pred solvePuzzle[] {  
  last[].far = Object  
}  
  
run solvePuzzle for 8 State
```


Alloy como Constraint Solver

Cambios de estado (cont.)

- Usando un predicado que especifica el estado final deseado, podemos utilizar Alloy para encontrar la solución:

Notar el uso de **last** para identificar cómo debería ser el estado final.

```
pred solvePuzzle[] {  
  last[].far = Object  
}
```

```
run solvePuzzle for 8 State
```

Modelos de Ejecuciones en Alloy

- Alloy es un lenguaje adecuado para la especificación de aspectos estáticos de software.
- Sin embargo, presenta una alternativa para describir (o analizar) aspectos dinámicos del mismo, es decir, propiedades de ejecuciones.
- Aún así, la idea presentada anteriormente sobre cómo modelar trazas puede aprovecharse para modelar y analizar propiedades dinámicas de sistemas.

Modelos de Ejecuciones en Alloy

La técnica para modelar ejecuciones y sus propiedades en Alloy consta de los siguientes pasos:

1. identificar la parte de la especificación que constituye el estado del sistema (usualmente será un conjunto de firmas).
2. utilizar esto para definir la firma del **estado**.
3. definir una **traza** como un orden total sobre estados.
4. indicar cuáles son las propiedades que se asumen del **estado inicial** con un hecho que predique sobre el primer elemento de las trazas.
5. indicar cómo se pasa de un estado a otro mediante hechos, que relacionen pares consecutivos en las trazas (define una **transición**)
6. usar aserciones para especificar las **propiedades** requeridas de las ejecuciones (como propiedades de los estados involucrados en las trazas).

Modelos de Ejecuciones en Alloy

Ejecuciones de Operaciones de Memorias

Consideremos un modelo simple de memorias:

```
sig Memory{
    data: Addr -> lone Data
}

pred init [ m: Memory ] { no m.data }

pred write [ m_i, m_o: Memory, a: Addr, d: Data ] {
    m_o.data = m_i.data ++ a -> d
}

pred read [ m: Memory, a: Addr, d_o: Data ] {
    let d = m.data[a] | some d implies d = d_o
}
```

Modelos de Ejecuciones en Alloy

Ejecuciones de Operaciones de Memorias (cont.)

Obviamente, Memory representa nuestro estado, así que la definición de trazas será la siguiente:

```
open util/ordering [ State ]

sig State {
    mem:Memory
}

fact Traces {
    init[first[] .mem]
    all s1:State, s2:next[s1] |
        (some a:Addr, d:Data | read[s1.mem,a,d] and s1 = s2)
        or
        (some a:Addr, d:Data | write[s1.mem,s2.mem,a,d])
}
```

Modelos de Ejecuciones en Alloy

Ejecuciones de Operaciones de Memorias (cont.)

Finalmente podemos expresar propiedades usando aserciones o predicados, de acuerdo a lo que busquemos:

```
pred freshDir [ m: Memory ] {  
    some a: Addr | no (m.data.Data & a)  
}  
  
pred freshDirInLast {  
    freshDir [last[].mem]  
}  
  
run freshDirInLast for 5
```

Uso de Alloy para verificar refinamientos

- En la práctica, muchas veces se trabaja con diferentes modelos de un mismo problema (o sistema).
- Estos difieren en el grado de abstracción empleado:
 - los más **abstractos** corresponden a la especificación,
 - los más **concretos** agregan más detalles de implementación
- Por ejemplo, el modelo de memorias con cache, puede verse como una versión más detallada, más concreta, más compleja, del modelo simple de memorias de las transparencias anteriores.
- Es decir, las memorias con cache son un **refinamiento** de las memorias simples.
- Sería interesante y útil poder **comparar** los diferentes modelos.
- Esto permitiría, por ejemplo, garantizar que las operaciones sobre las memorias más complejas se corresponden (i.e., respetan la especificación) de las memorias más abstractas.

Uso de Alloy para verificar refinamientos

- Tal comparación de modelos no puede hacerse directamente, pues estos definen espacios de estados diferentes.
- Una forma de relacionar modelos en diferentes niveles de abstracción es mediante **funciones de abstracción**.
- Una función de abstracción es una función que relaciona elementos de un modelo concreto con elementos de uno abstracto, mapeando cada estado concreto al estado abstracto que éste representa.
- En el ejemplo de los modelos de memoria cache y memoria abstracta, el map de memorias que una memoria con cache representa es la memoria principal "pisada" con el contenido de la cache:

```
fun alpha [c: CacheSystem]: Memory {  
    { m: Memory | m.data = c.main.map ++ c.cache.map }  
}
```


Uso de Alloy para verificar refinamientos

Con esta definición Alloy de la función de abstracción podemos:

- comprobar que no hay “pérdida” en la representación (la función de abstracción es suryectiva)
- comprobar que las operaciones concretas “implementan” las operaciones abstractas correspondientes:

```
assert WriteOK {  
    all c1, c2: CacheSystem,  
        a: Addr, d: Data, m1, m2: Memory |  
        ( cache/write[c1,c2,a,d] and  
          m1 = alpha[c1] and m2 = alpha[c2] )  
    =>  
        amemory/write[m1,m2,a,d]  
}
```

Análisis de Especificaciones en Alloy

El análisis nos ayuda a construir especificaciones de 3 maneras distintas:

- **Fomenta la exploración** a través de ejemplos concretos que confirman nuestra intuición y sugieren nuevos escenarios.
- **Supervisa**, ayudándonos a verificar sobre las versiones preliminares que lo que escribamos signifique lo que pretendemos significar.
- **Revela defectos** de manera temprana que de otra manera se hubieran revelado mucho más tarde (inclusive, ya entregado y en funcionamiento)

Análisis de Especificaciones en Alloy

Hemos visto que las especificaciones Alloy son analizables automáticamente usando técnicas de SAT solving.

Esencialmente, el uso de la técnica requiere:

- la caracterización de la propiedad a validar usando **predicados** o **aserciones**;
- la provisión de **cotas** adecuadas para los dominios de la especificación; y
- la interpretación de los **resultados** para la depuración de las especificaciones.

Análisis de Especificaciones en Alloy

Análisis de restricciones con Alloy Analyzer

- Para el caso de análisis de **predicados**:
 - La restricción se construye como la conjunción del predicado bajo estudio con todos los hechos del modelo (los explícitos y los implícitos en las declaraciones).
 - Si es satisfactible se genera un **ejemplo**, i.e., una instancia (o modelo) que **satisface** el predicado.
- Para el caso de análisis de **aserciones**:
 - La restricción se construye como la conjunción de la **negación** de la aserción bajo estudio con todos los hechos del modelo.
 - Si no es satisfactible se genera un **contraejemplo**, i.e., una instancia (o modelo) que hace que la aserción sea **falsa**.

Análisis de Especificaciones en Alloy

Cotas adecuadas en la búsqueda de instancias

- El éxito en el uso del Alloy Analyzer para validar propiedades de especificaciones **depende** en gran medida de la elección de **cotas** adecuadas para la búsqueda de instancias de aserciones o predicados.
- Al elegir cotas debemos ser cuidadosos:
cotas incorrectas pueden llevarnos a conclusiones erróneas.

Análisis de Especificaciones en Alloy

Sugerencias para la elección de cotas

- Asegurarse de que la cota de cada signatura es lo suficientemente grande como para permitir la existencia de constantes definidas en el modelo.

Por ejemplo:

```
some disj start, end: Node | ...
```

requiere como mínimo **dos** elementos en **Node**.

Análisis de Especificaciones en Alloy

Sugerencias para la elección de cotas (cont.)

- Si los elementos relevantes de una signatura se declaran como variables, no hace falta considerar cotas superiores al número de variables usadas.

Por ejemplo:

```
sig Name, Address {}
sig Alias, Group extends Name {}
sig Book {
    addr: Name -> Address
}
pred add [ b_i, b_o: Book, n: Name, a: Address ] {
    b_o.addr = b_i.addr + n->a
}
run add for 3 but 2 Book
```

Análisis de Especificaciones en Alloy

Sugerencias para la elección de cotas (cont.)

- Si los elementos relevantes de una signature se declaran como variables, no hace falta considerar cotas superiores al número de variables usadas.

Por ejemplo:

```
sig Name, Address {}
sig Alias, Group extends Name {}
sig Book {
  addr: Name -> Address
}

pred add [ b_i, b_o: Book, n: Name, a: Address ] {
  b_o.addr = b_i.addr + n->a
}

run add for 3 but 2 Book
```

Sólo se necesitan 2 **Books** para visualizar el predicado **add**

Análisis de Especificaciones en Alloy

Sugerencias para la elección de cotas (cont.)

- Por otro lado, el análisis del siguiente assert

```
sig Node {  
    adj: Node  
}  
assert allConectedGraphsAreClique {  
    all n1, n2: Node |  
        n2 in n1.*adj => (n2 in n1.adj or n2 = n1 )  
}  
check allConectedGraphsAreClique for 3
```

no debería limitarse a 2 **Nodes** porque las variables **n** y **n_** están cuantificadas universalmente sobre todos los **Nodes**

Análisis de Especificaciones en Alloy

Sugerencias para la elección de cotas (cont.)

- Por otro lado, el análisis del siguiente assert

```
sig Node {  
    adj: Node  
}  
assert allConectedGraphsAreClique {  
    all n1, n2: Node |  
        n2 in n1.*adj => (n2 in n1.adj or n2 = n1 )  
}  
check allConectedGraphsAreClique for 3
```

Observar que el assert es válido si la cota es 2

no debería limitarse a 2 **Nodes** porque las variables **n** y **n_** están cuantificadas universalmente sobre todos los **Nodes**

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados

- Un problema que puede dar lugar a contraejemplos no intuitivos es el que surge al querer representar, mediante una signatura, **todos** los valores posibles de una estructura compuesta.
- Esto va, en un sentido, en contra de la semántica de Alloy, que interpreta a las signaturas **sólo** como **un conjunto** de elementos.
- Esta contradicción sólo se manifiesta cuando se utiliza cuantificación universal de cierta forma (concretamente, de manera no acotada).

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Consideremos el siguiente ejemplo:

```
sig Element {}

sig Set {
  elements: set Element
}

assert Closed {
  all s0, s1: Set | some s2: Set |
    s2.elements = s0.elements + s1.elements
}

check Closed
```

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Esta aserción nos puede dar contraejemplos tales como el siguiente:

```
Set = { (S0), (S1) }  
Element = { (E0), (E1) }  
s0 = { (S0) }  
s1 = { (S1) }  
elements = { (S0, E0), (S1, E1) }
```

??!!

¿Dónde está el problema?

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

El analizador encontró un contraejemplo que no “completó” la signatura **Set** con valores suficientes: no existen instancias de **Set** con el conjunto vacío, ni con los elementos **E0** y **E1**.

Dado que queremos que haya tantas instancias de Set como subconjuntos de Element, podemos forzar esto mediante axiomas de generación.

```
fact SetGenerator {  
    some s: Set | no s.elements  
    all s1: Set, e: Element |  
        some s2: Set | s2.elements = s1.elements + e  
}
```

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

El analizador encontró un contraejemplo que no "completó" la
signatura **Set** con valores suficientes. El ejemplo encontrado
con el conjunto vacío, ni con un solo elemento.

Dado que queremos que **Set** sea un conjunto de
subconjuntos de **Element**, necesitamos un axioma de generación.

¿Qué sucederá si analizamos la
aserción anterior, teniendo en cuenta el
axioma de generación?

El resultado es el
esperado:

"No counterexample found.
Assertion may be valid."

```
fact SetGenerator {  
    some s: Set | no s.elements  
    all s1: Set, e: Element |  
        some s2: Set | s2.elements = s1.elements + e  
}
```

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Notar: las interpretaciones que el AlloyAnalyzer explorará son **finitas**

Por lo tanto, si el axioma de generación fuerza infinitos átomos, la especificación asociada se hará inconsistente en el análisis.

```
abstract sig List {}
one sig EmptyList extends List {}
sig NonEmptyList extends List {
    element: Element,
    rest: List
}
fact ListGenerator {
    all l1: List, e: Element |
        some l2: List | l2.rest = l1 and l2.element = e
}
```


Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Sobre el ejemplo anterior:

```
assert ObviouslyNotValid {  
    all l1: List |  
        all l2: List | l2.rest = l1 or l1 = l2  
}  
check ObviouslyNotValid
```

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Sobre el ejemplo anterior:

```
assert ObviouslyNotValid {  
    all l1: List |  
        all l2: List | l2.rest = l1 or l1 = l2  
}  
check ObviouslyNotValid
```

??!!

Alloy Analyzer no encuentra
contraejemplo para esta
aserción

Es consecuencia
del axioma de generación
forzando infinitas listas

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Sobre el ejemplo anterior:

¿Cuántas instancias de lista se muestra al correr lo siguiente?

```
pred show [] {}  
run show
```

¿Y en el siguiente caso?

```
pred show [] {}  
run show for 3 but exactly 1 Element
```

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

Sobre el ejemplo anterior:

¿Cuántas instancias de lista se muestra al correr lo siguiente?

```
pred show [] {}  
run show
```

1

¿Por qué?

¿Y en el siguiente caso?

```
pred show [] {}  
run show for 3 but exactly 1 Element
```

0

Análisis de Especificaciones en Alloy

Cuantificadores universales no acotados (cont.)

- Cabe aclarar que la necesidad del axioma de generación surge como consecuencia de la cuantificación universal no acotada.
- Existen ciertas formas en las aserciones para las cuales está garantizado que no es necesario introducir axiomas de generación.
- Éstas son las expresiones con cuantificación universal acotada.
- Afortunadamente, son las más comunes en la práctica.